```
(*regularized distance*)
re[r_, e_] := Sqrt[r^2 + e^2];

(*
Since the regularized Kelvinlet solutions are radially symmetric,
the asymptotic analysis can be computed as a function of the radius r.
  Also, note that the expansion is performed in terms of the Kelvinlet matrix,
instead of on the displacement field u, so it is independent of the force values
*)

(*3D regularized Kelvinlet*)
Kelvinlet3[r_, e_] :=
 (a - b) / re[r, e] + (a / 2) * e^2 / re[r, e]^3 + b * r^2 / re[r, e]^3
Kelvinlet3[r, e]
```

$$\frac{a\,e^2}{2\left(e^2+r^2\right)^{3/2}}+\frac{b\,r^2}{\left(e^2+r^2\right)^{3/2}}+\frac{a-b}{\sqrt{e^2+r^2}}$$

```
FullSimplify[Series[Kelvinlet3[r, e], {r, Infinity, 6}]]
```

$$\frac{a}{r}-\frac{b\,e^2}{r^3}-\frac{3\left(\left(a-4\,b\right)\,e^4\right)}{8\,r^5}+O\left[\frac{1}{r}\right]^7$$

```
(*3D locally affine regularized Kelvinlets*)
Affine3[r_, e_] :=
  (a - b) * r / re[r, e]^3 + (a / 2) * r * e^2 / re[r, e]^5 + b * r^3 / re[r, e]^5;
Affine3[
 r,
 e]
```

$$\frac{a\,e^2\,r}{2\left(e^2+r^2\right)^{5/2}}+\frac{b\,r^3}{\left(e^2+r^2\right)^{5/2}}+\frac{\left(a-b\right)\,r}{\left(e^2+r^2\right)^{3/2}}$$

```
FullSimplify[Series[Affine3[r, e], {r, Infinity, 7}]]
```

$$\frac{a}{r^2}-\frac{\left(a+b\right)\,e^2}{r^4}+\frac{5\left(a+4\,b\right)\,e^4}{8\,r^6}+O\left[\frac{1}{r}\right]^8$$

```
(*2D regularized Kelvinlet*)
Kelvinlet2[r_, e_] :=
  2 * (a - b) * Log[1 / re[r, e]] + a * e^2 / re[r, e]^2 + 2 * b * r^2 / re[r, e]^2;
Kelvinlet2[
 r,
 e]
```

$$\frac{a\,e^2}{e^2+r^2}+\frac{2\,b\,r^2}{e^2+r^2}+2\left(a-b\right)\,\mathrm{Log}\left[\frac{1}{\sqrt{e^2+r^2}}\right]$$

```
FullSimplify[Series[Kelvinlet2[r, e], {r, Infinity, 6}]]
```

$$2\left(b+\left(a-b\right)\,\mathrm{Log}\left[\frac{1}{r}\right]\right)-\frac{b\,e^2}{r^2}-\frac{\left(a-3\,b\right)\,e^4}{2\,r^4}+\frac{\left(2\,a-5\,b\right)\,e^6}{3\,r^6}+O\left[\frac{1}{r}\right]^7$$

```
(*2D locally affine regularized Kelvinlets*)
Affine2[r_, e_] :=
  2 * (b - a) * r / re[r, e] ^2 - 2 * a * r * e^2 / re[r, e] ^4 - 4 * b * r^3 / re[r, e] ^4;
Affine2[
 r,
 e]
```

$$-\frac{2\,a\,e^2\,r}{\left(e^2+r^2\right)^2}-\frac{4\,b\,r^3}{\left(e^2+r^2\right)^2}+\frac{2\,(-a+b)\,r}{e^2+r^2}$$

```
FullSimplify[Series[Affine2[r, e], {r, Infinity, 5}]]
```

$$-\frac{2\,(a+b)}{r}+\frac{6\,b\,e^2}{r^3}+\frac{2\,(a-5\,b)\,e^4}{r^5}+O\left[\frac{1}{r}\right]^6$$